Appendix A

SPNE derivation

assumptions

```
A0 = And[t > 0, g ≥ 0, r > 0, c1 ≥ 0, c2 ≥ 0, a1 ≥ 0, a2 ≥ 0, n ≥ 1, v1 ≥ 0, v2 ≥ 0];
A1 = (4 (v1 + v2) > 5 (c1 + c2) + 24 t);
A2 = (- r (3 g + 2 r) / (6 n) + 6 t ≥ c2 - c1 + v1 - v2 ≥ - r (r (-2 + 1 / n) + 3 g / n) / 6 - 6 t);
A3 = (r^2 < 24 t);
A4 = (v1 - c1 == v2 - c2);
A5 = (g < (-3 + 2 n) r / 6);
A6 = (v1 - v2 > (5 (6 g + (3 - 2 n) r)^2 t) / ((1 + 2 n) r (12 g + (3 - 2 n) r) + 72 (3 - 2 n) n t));
```

weaker variants of A0 and A3

```
A0b = And[t > 0, g ≥ 0, r > 0, c1 ≥ 0, c2 ≥ 0, a1 ≥ 0, a2 ≥ 0, n > 0, v1 ≥ 0, v2 ≥ 0];
A3b = ((1 + 2 n) r^2 - 72 n t < 0);
Simplify[Implies[A3, A3b], A0]
Simplify[Implies[A0, A0b]]
```

```
True

True
```

define utility functions (skip the retailer index j for convenience)

```
V1 = v1 + r * a1 + g * (a1 + a2); V2 = v2 + r * a2 + g * (a1 + a2);
u1 = V1 - p1 - t * x;
u2 = V2 - p2 - t * (1 - x);
```

indifferent consumer locations (market covered / uncovered)

```
xCov = Solve[u1 == u2, x][[1, 1, 2]];
xUn1 = Solve[u1 == 0, x][[1, 1, 2]]; xUn2 = Solve[u2 == 0, x][[1, 1, 2]];
```

profit-maximizing retail prices when market not covered

```
ProfRUnc = (1 / n) ((p1 - w1) * xUn1 + (p2 - w2) (1 - xUn2)) - a1^2;
NEpUnc = FullSimplify[
  Join[Solve[D[ProfRUnc, p1] == 0, p1][[1]], Solve[D[ProfRUnc, p2] == 0, p2][[1]]]]
```

$$\left\{p1 \to \frac{1}{2}\left(a2\, g + a1\,(g + r) + v1 + w1\right),\ p2 \to \frac{1}{2}\left(a1\, g + a2\,(g + r) + v2 + w2\right)\right\}$$

profit-maximizing retail prices when market covered

```
ProfRCov = (1/n) ((p1 - w1) * xCov + (p2 - w2) (1 - xCov)) - a1^2;
MaxP2 = Solve[(u1 /. x -> xCov) == 0, p2][[1]];
ProfRCovOneVar = ProfRCov /. MaxP2;
NEpCov = Solve[D[ProfRCovOneVar, p1] == 0, p1][[1]];
NEpCov = FullSimplify[Join[NEpCov, MaxP2 /. NEpCov]]
```

$$\left\{p1 \to \frac{1}{4}\left(4\,(a1 + a2)\,g + (3\,a1 + a2)\,r - 2\,t + 3\,v1 + v2 + w1 - w2\right),\ p2 \to \frac{1}{4}\left(4\,(a1 + a2)\,g + (a1 + 3\,a2)\,r - 2\,t + v1 + 3\,v2 - w1 + w2\right)\right\}$$

reduced game equilibrium wholesale prices when market not covered

```
ProfM1Unc = (1/n) ((w1 - c1) * xUn1) /. NEpUnc;
ProfM2Unc = (((w2 - c2) * (1 - xUn2)) /. NEpUnc) - a2^2;
NEwUnc = Solve[D[ProfM1Unc, w1] == D[ProfM2Unc, w2] == 0, {w1, w2}][[1]]
```

$$\left\{w1 \to \frac{1}{2}\left(c1 + a1\,g + a2\,g + a1\,r + v1\right),\ w2 \to \frac{1}{2}\left(c2 + a1\,g + a2\,g + a2\,r + v2\right)\right\}$$

reduced game equilibrium wholesale prices when market covered

```
ProfM1Cov = (1/n) ((w1 - c1) * xCov);
ProfM2Cov = (((w2 - c2) * (1 - xCov))) - a2^2;
NEwCov =
 Solve[D[(ProfM1Cov /. NEpCov), w1] == D[(ProfM2Cov /. NEpCov), w2] == 0, {w1, w2}][[1]]
```

$$\left\{w1 \to \frac{1}{3}\left(2\,c1 + c2 + a1\,r - a2\,r + 6\,t + v1 - v2\right),\ w2 \to \frac{1}{3}\left(c1 + 2\,c2 - a1\,r + a2\,r + 6\,t - v1 + v2\right)\right\}$$

uncovered market possible?

```
Reduce[
 And[Or[((xUn1 < xUn2) /. NEpUnc /. NEwUnc), ((xUn1 < xUn2) /. NEpUnc /. NEwCov)], A0, A1]]
False
```

SB vs. NB

reduced game equilibrium advertising when market covered

profit functions concave?

```
ProfRStage1 = (ProfRCov /. NEpCov /. NEwCov);
ProfM2Stage1 = (ProfM2Cov /. NEpCov /. NEwCov);
Simplify[And[D[ProfRStage1, {a1, 2}] < 0], And[A0, A1, A2, A3b]]
True
```

equilibrium advertising

```
NEa = FullSimplify[Solve[D[ProfRStage1, a1] == D[ProfM2Stage1, a2] == 0, {a1, a2}][[1]]]
```

$$\left\{a1 \to \frac{3\,c1\,r - 3\,c2\,r + 3\,g\,r^2 + 2\,r^3 - 108\,g\,t - 54\,r\,t - 3\,r\,v1 + 3\,r\,v2}{3\,r^2 + 6\,n\,r^2 - 216\,n\,t},\; a2 \to \frac{r\,(r\,(3\,g + 2\,r) - 6\,n\,(c1 - c2 + 6\,t - v1 + v2))}{3\,((1 + 2\,n)\,r^2 - 72\,n\,t)}\right\}$$

corner solution (or negative advertising) possible?

```
xNE = FullSimplify[(xCov /. NEpCov /. NEwCov /. NEa)];
Reduce[And[Not[0 ≤ xNE ≤ 1], A0, A1, A2, A3b], Reals]
Reduce[And[Or[NEa[[1, 2]] < 0, NEa[[2, 2]] < 0], And[A0b, A1, A2, A3b]]]
False

False
```

NB vs NB

reduced game equilibrium advertising when market covered

profit functions concave?

```
ProfM1Stage1 = ((((w1 - c1) * xCov) - a1^2) /. NEpCov /. NEwCov);
ProfM2Stage1 = (ProfM2Cov /. NEpCov /. NEwCov);
Simplify[And[D[ProfM1Stage1, {a1, 2}] < 0], And[A0, A1, A2, A3b]]
Simplify[And[D[ProfM2Stage1, {a2, 2}] < 0], And[A0, A1, A2, A3b]]
True

True
```

equilibrium advertising

```
NEa2NB =
 FullSimplify[Solve[D[ProfM1Stage1, a1] == D[ProfM2Stage1, a2] == 0, {a1, a2}][[1]]]
```

$$\left\{a1 \to \frac{r\,(3\,c1 - 3\,c2 + r^2 - 18\,t - 3\,v1 + 3\,v2)}{6\,(r^2 - 18\,t)},\; a2 \to \frac{r\,(-3\,c1 + 3\,c2 + r^2 - 18\,t + 3\,v1 - 3\,v2)}{6\,(r^2 - 18\,t)}\right\}$$

corner solution (or negative advertising) possible?

```
xNE2NB = FullSimplify[(xCov /. NEpCov /. NEwCov /. NEa2NB)];
Reduce[And[Not[0 ≤ xNE2NB ≤ 1], A0, A1, A2, A3, A4], Reals]
Reduce[And[Or[NEa2NB[[1, 2]] < 0, NEa2NB[[2, 2]] < 0], And[A0b, A1, A2, A3, A4]]]
False

False
```

Results

Proposition 1

```
pDne = Simplify[((p1 - p2) /. NEpCov /. NEwCov /. NEa)];
wDne = Simplify[((w1 - w2) /. NEwCov /. NEa)];
aDne = Simplify[((a1 - a2) /. NEa)];
```

```
Simplify[And[D[aDne, c1] < 0, D[xNE, c1] < 0], And[A0, A1, A2, A3]]
```

True

```
Simplify[And[D[aDne, c2] > 0, D[xNE, c2] > 0], And[A0, A1, A2, A3]]
```

True

```
Simplify[And[D[pDne, g] > 0, D[wDne, g] > 0, D[aDne, g] > 0, D[xNE, g] > 0],
 And[A0, A1, A2, A3]]
```

True

```
Reduce[Not[And[D[pDne, n] < 0, D[wDne, n] < 0, D[aDne, n] < 0, D[xNE, n] < 0]] &&
  And[A0b, A1, A2, A3b]]
```

False

```
Simplify[And[D[pDne, v1] > 0, D[wDne, v1] > 0, D[aDne, v1] > 0, D[xNE, v1] > 0],
 And[A0, A1, A2, A3]]
```

True

```
Simplify[And[D[pDne, v2] < 0, D[wDne, v2] < 0, D[aDne, v2] < 0, D[xNE, v2] < 0],
 And[A0, A1, A2, A3]]
```

True

Effect of t on pDne, wDne, aDne and xNE has the same direction under A0 - A3 (prove negation is false)

```
Reduce[Not[Equivalent[D[pDne, t] < 0, D[wDne, t] < 0, D[aDne, t] < 0, D[xNE, t] < 0]] &&
  And[A0, A1, A2, A3]]
```

False

Proposition 2 (prove negation is false)

```
Reduce[
 Not[And[D[pDne, r] < 0, D[wDne, r] < 0, D[aDne, r] < 0, D[xNE, r] < 0, D[pDne, t] > 0,
    D[wDne, t] > 0, D[aDne, t] > 0, D[xNE, t] > 0]] && And[A0, A1, A2, A3, A4, A5]]
```

False

Under A4 xNE is independent of innate valuations and costs

```
FullSimplify[xNE, A4]
```

$$-\frac{r\left(3g+r-2nr\right)+36nt}{\left(1+2n\right)r^2-72nt}$$

Proposition 3 (prove negation is false)

```
R = FullSimplify[pDne / aDne, A4];
Reduce[Not[Equivalent[A5, aDne < 0]] && And[A0b, A1, A2, A3, A4]]
```

```
False
```

```
Reduce[And[D[R, r] < 0, A0b, A1, A2, A3, A4, A6, aDne < 0], Reals]
```

```
False
```

```
Reduce[And[Not[Implies[v1 - v2 > 0, A6]], A0, A1, A2, A3, A4, A5]]
```

```
False
```

Welfare - SB vs NB

```
CSwithADV =
  ((Integrate[V1 - p1 - t * x, {x, 0, xInd}] + Integrate[V2 - p2 - t * (1 - x), {x, xInd, 1}]) /.
      NEpCov /. NEwCov /. NEa /. xInd → xNE);
CSnoAdv = ((Integrate[V1 - p1 - t * x, {x, 0, xInd}] + Integrate[V2 - p2 - t * (1 - x),
          {x, xInd, 1}]) /. xInd → xCov) /. NEpCov /. NEwCov /. {a1 → 0, a2 → 0};
Reduce[CSwithADV < CSnoAdv && And[A0, A1, A2, A3, A4]]
```

```
False
```

```
SWadv = CSwithADV + PSwithAdv; SWnoadv = CSnoAdv + PSnoAdv;
Reduce[Not[SWadv > SWnoadv] && And[A0, A1, A2, A3, A4]]
```

```
False
```

```
(*Reduce[D[SWnoadv-SWadv,r]<0&&And[A0,A1,A2,A3,A4]];*)
```

```
SNBV = {{CSwithADV, CSnoAdv}, {PSwithAdv, PSnoAdv}, {SWadv, SWnoadv}};
```

Welfare - NB vs NB

```
CSwithADV =
  ((Integrate[V1 - p1 - t * x, {x, 0, xInd}] + Integrate[V2 - p2 - t * (1 - x), {x, xInd, 1}]) /.
      NEpCov /. NEwCov /. NEa2NB /. xInd → xNE2NB);
CSnoAdv = ((Integrate[V1 - p1 - t * x, {x, 0, xInd}] + Integrate[V2 - p2 - t * (1 - x),
          {x, xInd, 1}]) /. xInd → xCov) /. NEpCov /. NEwCov /. {a1 → 0, a2 → 0};
Reduce[CSwithADV ≠ CSnoAdv && And[A0, A1, A2, A3, A4]]
```

```
False
```

```
SWadv = CSwithADV + PSwithAdv; SWnoadv = CSnoAdv + PSnoAdv;
Reduce[Not[SWadv > SWnoadv] && And[A0, A1, A2, A3, A4]]
```

```
False
```

```
Reduce[Not[D[SWnoadv - SWadv, r] < 0] && And[A0, A1, A2, A3, A4]]
```

```
False
```

```
NNBV = {{CSwithADV, CSnoAdv}, {PSwithAdv, PSnoAdv}, {SWadv, SWnoadv}};
```

Proposition 4

```
(*consumers gain from adv., but only if SB present*)
```

```
Reduce[Not[SNBV[[1, 1]] ≥ NNBV[[1, 1]] == NNBV[[1, 2]] == SNBV[[1, 2]]] &&
  And[A0, A1, A2, A3, A4]]
```

```
False
```

```
(*producers gain from adv., but (when n small) more so when SB present*)
```

```
Reduce[Not[NNBV[[2, 1]] > NNBV[[2, 2]]] && And[A0, A1, A2, A3, A4]]
```

nsmall = $\left(n < \frac{1}{2} - \sqrt{\frac{\left(r^2 - 27\,t\right)^2 \left(2\,r^4 - 45\,r^2\,t + 729\,t^2\right)}{\left(2\,r^4 - 135\,r^2\,t + 1296\,t^2\right)^2} + \frac{9\,t\left(-4\,r^2 + 9\,t\right)}{2\,r^4 - 135\,r^2\,t + 1296\,t^2}}\right)$;

```
Reduce[Not[SNBV[[2, 1]] > NNBV[[2, 1]]] && And[A0, A1, A2, A3, A4] && nsmall]
```

```
False
```

```
False
```

```
(*r increases consumer surplus with store
  brand when either n small or n big so that a1 < a2*)
```

```
Reduce[D[SNBV[[1, 1]], r] ≤ 0 &&
  And[A0, A1, A2, A3, A4] && Or[NEa[[1, 2]] - NEa[[2, 2]] < 0, nsmall]]
```

```
False
```

```
(*r increases producer surplus with store
  brand when either n small or n big so that a1 < a2 *)
```

```
Reduce[D[SNBV[[2, 1]], r] ≤ 0 &&
  And[A0, A1, A2, A3, A4] && Or[NEa[[1, 2]] - NEa[[2, 2]] < 0, nsmall]]
```

```
False
```